

9/6.4

Création de jeux d'aventures

Le programme présenté ici permet de créer de façon simple des jeux d'aventures dans lesquels il y aura plusieurs tableaux (aucune limite quant à leur nombre, si ce n'est la place mémoire disponible), et où le joueur pourra dialoguer avec l'ordinateur pour ramasser des objets, avancer dans une direction ou effectuer une autre action quelconque (les diverses actions possibles sont définies dans ce programme).

Une analyse syntaxique de mots-clés du type « VERBE SUJET » (voir Partie 9, chapitre 6.1) est faite et permet de définir l'action à effectuer en fonction du couple de mots-clés identifiés.

Les points de vie sont incrémentés ou décrémentés en fonction des actions du joueur.

Une musique peut éventuellement être exécutée sous interruptions en parallèle avec le jeu (voir Partie 6, chapitre 5.1).

Enfin, une fenêtre déroulante peut permettre au joueur de visualiser quels sont les mots-clés compris par la machine (Voir Partie 9, chapitre 6.2).

Utilisation du programme

Dès le début du programme, un menu est affiché sur l'écran. Il comporte les options suivantes :

- charger une aventure existante ;
- modifier ou créer le nom des salles de jeu ;
- modifier ou créer les mots-clés (Verbes et Sujets) du jeu ;
- modifier ou créer les réponses que donnera l'ordinateur suite aux actions du jour ;
- définir les actions découlant de l'association SALLE-VERBE-SUJET décryptée par l'analyseur de syntaxe ;
- créer le fichier Aventure exécutable par le programme d'exécution d'aventure défini à la Partie 9, chapitre 6.4.

Lors de la définition des actions, 5^e option du menu ci-dessus, il vous faudra entrer :

- le numéro de la salle ;
- le numéro du verbe ;
- le numéro du sujet ;
- le code action correspondant.

Ce code peut comporter une association quelconque des quatre commandes suivantes :


- changement de salle (*Exemple* : C4 fait passer en salle 4) ;
- réponse de l'ordinateur (*Exemple* : R2 fait afficher la réponse numéro 2) ;
- modification des points du joueur (*Exemple* : P10 fait augmenter le nombre de points du joueur de 10, et P-20 fait diminuer de 20 ce même nombre) ;
- fin du jeu. La partie est terminée (F).

Remarques :

- maximum de 80 caractères,
- écrits de la façon suivante : Lettre,O

Par exemple, l'action changement de salle, passage en salle 4 et fin de jeu sera codée : C4ROPOF.

La logique du programme est la suivante :



Le listing du programme est le suivant :

```

1000 MEMORY &5F00
1010 REM CREATION D'AVENTURES DE TYPE 1
1020 DIM LAB$(100,4),N(4)
1030 '=====
1040 REM Menu
1050 '
1060 INK 0,0:BORDER 0:INK 1,10:INK 2,3:INK 3,1,3:MODE 1:PEN
3
1070 PRINT"      CREATION D'AVENTURES TYPE 1":PEN 1
1080 LOCATE 13,8:PRINT"1- Chargement"
1090 LOCATE 13,10:PRINT"2- Salles"
1100 LOCATE 13,12:PRINT"3- Mots-cles"
1110 LOCATE 13,14:PRINT"4- Repónses"
1120 LOCATE 13,16:PRINT"5- Conditions"
1130 LOCATE 13,18:PRINT"6- Fin"
1140 A$=INKEY$:IF A$="" THEN 1140
1150 A=ASC(A$)
1160 IF A>54 OR A<49 THEN SOUND 1,100,30 :GOTO 1140
1170 ON A-48 GOSUB 2000,3000,4000,6000,5000,7000
1180 GOTO 1060 'Boucle sur menu
2000 '=====
2010 'CHARGEMENT D'UNE AVENTURE
2020 '
2030 CLS:PEN 3:PRINT"  CHARGEMENT D'UNE AVENTURE":PEN 1
2040 WINDOW 1,40,5,25
2050 INPUT"Nom de l'aventure ";N$
2060 LOAD N$,&6000 'Chargement
2070 AD=&6000 'Adresse de depart du fichier aventure
2080 SD=PEEK(AD):AD=AD+2 'Salle de depart
2090 IF PEEK(AD)=&FF THEN T$="":AD=AD+1 ELSE GOSUB 2320:T$=X
$
2100 IF PEEK(AD)=&FF THEN MUS$="":AD=AD+1 ELSE GOSUB 2320:MU
S$=X$
2110 HELP=PEEK(AD):AD=AD+2
2120 FOR I=1 TO 4
2130   N(I)=PEEK(AD):AD=AD+1
2140   FOR J=1 TO N(I)
2150     GOSUB 2320:LAB$(J,I)=X$
2160   NEXT J
2170 NEXT I
2180 NR=PEEK(AD)
2190 FOR I=1 TO NR
2200   FOR J=1 TO 3
2210     AD=AD+1:CAR(I,J)=PEEK(AD)
2220   NEXT J
2230   X$="":AD=AD+1:P=PEEK(AD):IF P<>0 THEN X$=X$+"C"+RIGHT
$(STR$(P),LEN(STR$(P))-1)
2240   AD=AD+1:P=PEEK(AD):IF P<>0 THEN X$=X$+"R"+RIGHT$(STR$
(P),LEN(STR$(P))-1)

```

```

2250 AD=AD+1:P=PEEK(AD):IF P=0 THEN 2280
2260 X$=X$+"P":LL=LEN(STR$(P-128))
2270 IF P-128<0 THEN X$=X$+RIGHT$(STR$(P-128),LL) ELSE X$=
X$+RIGHT$(STR$(P-128),LL-1)
2280 AD=AD+1:P=PEEK(AD):IF P<>0 THEN X$=X$+"F"
2290 C$(I)=X$
2300 NEXT I
2310 RETURN
2320 'Conversion ASCII -> Chaine de caracteres
2330 X$=""
2340 IF PEEK(AD)<>255 THEN X$=X$+CHR$(PEEK(AD)):AD=AD+1:GOTO
2340 ELSE AD=AD+1:RETURN
3000 '=====
3010 'ENCHAINEMENT DES PIECES
3020 '
3030 CLS:PEN 3:PRINT" ENCHAINEMENT DES PIECES":PEN 1:CA
=1
3040 WINDOW 1,40,5,25
3050 IF CA=1 THEN PR$="Donnez le nom des ECRANS"
3060 IF CA=2 THEN PR$="Donnez le nom des VERBES"
3070 IF CA=3 THEN PR$="Donnez le nom des SUJETS"
3080 IF CA=4 THEN PR$="Entrez les reponses"
3090 IF N(CA)<>0 THEN 3110 'Ecrans deja existant
3100 PRINT PR$;"et terminez la saisie par ENTER.":PRINT:GOTO
3530
3110 PEN 2:IF CA=1 THEN PRINT"ECRANS";
3120 IF CA=2 THEN PRINT"VERBES";
3130 IF CA=3 THEN PRINT"SUJETS";
3140 IF CA=4 THEN PRINT"REPONSES";
3150 PRINT" deja en memoire":PEN 1
3160 PRINT"Voulez-vous 1)Modifier"
3170 PRINT" 2)Ajouter"
3180 PRINT" 3)Supprimer des labels "
3190 A$=INKEY$:IF A$="" THEN 3190
3200 R=ASC(A$)-48
3210 IF ASC(A$)=13 THEN RETURN 'Retour au menu
3220 IF R>3 OR R<1 THEN SOUND 1,100,30:GOTO 3190
3230 ON R GOTO 3240,3330,3410
3240 'Modification de labels
3250 '
3260 PRINT:INPUT"L)iste,NO DE LABEL OU ENTER ";A$:PRINT
3270 IF LEN(A$)=0 THEN CLS:GOTO 3050
3280 IF A$="L" OR A$="1" THEN FOR I=1 TO N(CA):PRINT I;LAB$(
I,CA):NEXT:GOTO 3260
3290 A=VAL(A$):IF A=0 THEN SOUND 1,100,30:GOTO 3260
3300 IF A>N(CA) THEN SOUND 1,100,10:GOTO 3260
3310 PRINT:PRINT A;:INPUT A$:LAB$(A,CA)=A$
3320 GOTO 3260
3330 'Ajout de labels
3340 '
3350 PRINT:INPUT"L)iste OU ENTER ";A$:PRINT

```

```

3360 IF A$="L" OR A$="1" THEN FOR I=1 TO N(CA):PRINT I;LAB$(
I,CA):NEXT:GOTO 3350
3370 IF LEN(A$)<>0 THEN SOUND 1,100,30:GOTO 3350
3380 N(CA)=N(CA)+1:PRINT N(CA);:INPUT A$:LAB$(N(CA),CA)=A$
3390 IF LEN(A$)=0 THEN N(CA)=N(CA)-1:CLS:GOTO 3050
3400 GOTO 3380
3410 'Suppression de labels
3420 '
3430 PRINT:INPUT"L)iste,No label ou ENTER ";A$:PRINT
3440 IF LEN(A$)=0 THEN CLS:GOTO 3050
3450 IF A$="L" OR A$="1" THEN FOR I=1 TO N(CA):PRINT I;LAB$(
I,CA):NEXT:GOTO 3430
3460 A=VAL(A$):IF A=0 THEN SOUND 1,100,10:GOTO 3410
3470 IF A>N(CA) THEN SOUND 1,100,10:GOTO 3410
3480 PRINT A;LAB$(A,CA);" ";
3490 INPUT"OK (O/N) ";R$:R$=UPPER$(R$)
3500 IF R$<>"O" AND R$<>"N" THEN SOUND 1,100,10:GOTO 3490
3510 IF R$="O" THEN FOR I=A TO N(CA):LAB$(I,CA)=LAB$(I+1,CA)
:NEXT:N(CA)=N(CA)-1:GOTO 3260
3520 GOTO 3430
3530 'Saisie de labels
3540 '
3550 N(CA)=0 'Initialisation
3560 N(CA)=N(CA)+1:PRINT N(CA);:INPUT A$:LAB$(N(CA),CA)=A$
3570 IF LEN(A$)=0 THEN N(CA)=N(CA)-1:RETURN
3580 GOTO 3560
4000 '=====
4010 'DEFINITION DES MOTS-CLES
4020 '
4030 CLS:PEN 3:PRINT"          MOTS CLE":PEN 1
4040 WINDOW 1,40,5,25
4050 PRINT"V)erbes ou S)ujets "
4060 A$=INKEY$:IF A$="" THEN 4060
4070 A$=UPPER$(A$)
4080 IF A$<>"V" AND A$<>"S" THEN SOUND 1,100,30:GOTO 4060
4090 IF A$="V" THEN CA=2 ELSE CA=3
4100 GOTO 3040
5000 '=====
5010 'CONDITIONS ENTREES SALLES ORDRES ET MOTS-CLES
5020 '
5030 CLS:PEN 3:PRINT"  CONDITIONS LIANT SALLES ET MOTS-CLES"
:PEN 1
5040 WINDOW 1,40,5,25
5050 CLS:PRINT:PRINT"Voulez-vous 1)Lister les regles"
5060 PRINT"          2)Creer d'autres regles"
5070 PRINT"          3)Modifier une regle"
5080 A$=INKEY$:IF A$="" THEN 5080
5090 IF ASC(A$)=13 THEN RETURN
5100 A=VAL(A$):IF A<1 OR A>3 THEN SOUND 1,100,30:GOTO 5080
5110 ON A GOTO 5120,5230,5340
5120 'Listage des regles existantes
5130 IF NR=0 THEN PEN 3:LOCATE 10,10:PRINT"Aucune regle en m
emoire":PEN 1

```

```

5140 IF NR=0 THEN A$=INKEY$:IF A$="" THEN 5140
5150 FOR I=1 TO NR
5160   CLS:PRINT"Salle : ";LAB$(CAR(I,1),1)
5170   PRINT"Verbe : ";LAB$(CAR(I,2),2)
5180   PRINT"Sujet : ";LAB$(CAR(I,3),3)
5190   PRINT"Action: ";C$(I)
5200   A$=INKEY$:IF A$="" THEN 5200
5210 NEXT I
5220 GOTO 5050
5230 'Creation de regles
5240 NR=NR+1 'Nombre de relations+1
5250 LOCATE 1,7:INPUT "Salle ";CAR(NR,1)
5260 IF CAR(NR,1)=0 THEN NR=NR-1:GOTO 5050 'Sortie
5270 INPUT "Verbe ";CAR(NR,2)
5280 INPUT "Sujet ";CAR(NR,3)
5290 PRINT:PRINT:PRINT"(C=Changt salle,R=Reponse,P=Points)"
5300 PRINT"(F=Fin de partie)"
5310 PRINT"(Ex:C3P-10 =>Passage salle 3 et"
5320 PRINT"          nombre de points-10 )"
5330 LOCATE 1,10:INPUT"Action ";C$(NR):GOTO 5050
5340 'Modification d'une regle
5350 INPUT "No de la regle ";N
5351 IF N=0 THEN 5050
5360 IF N>NR THEN PRINT"Regle inexistante":GOTO 5200 'Retour
    au menu
5370 LOCATE 1,8:PRINT"Salle ";CAR(N,1)
5380 PRINT"Verbe ";CAR(N,2)
5390 PRINT"Sujet ";CAR(N,3)
5400 PRINT"Action ";C$(N)
5410 LOCATE 12,8 :INPUT CAR(N,1)
5420 LOCATE 12,9 :INPUT CAR(N,2)
5430 LOCATE 12,10:INPUT CAR(N,3)
5440 LOCATE 18,11:INPUT C$(N)
5450 GOTO 5050
6000 '=====
6010 'REPONSES DE LA MACHINE
6020 '
6030 CLS:PEN 3:PRINT"          REPONSES DE L'ORDINATEUR"
6040 CA=4:PEN 1:GOTO 3040
7000 '=====
7010 'SORTIE DU GENERATEUR D'AVENTURES
7020 '
7030 CLS:PEN 3:PRINT"          SAUEGARDE DE L'AVENTURE":PEN 1

7040 WINDOW 1,40,5,25
7050 CLS:PRINT:PRINT"Avant de sauvegarder le fichier aventur
erepondez aux questions suivantes:":PRINT
7060 PEN 2:PRINT"Salle de depart (1 a";N(1);") ";:PEN 1
7070 IF SD<>0 THEN PRINT SD
7080 LOCATE 25,5:INPUT SD$:PRINT
7090 IF SD$="" AND SD=0 THEN RETURN
7100 IF SD$="" THEN 7120 ELSE SD=VAL(SD$)

```

```

7110 IF SD<1 OR SD>N(1) THEN SOUND 1,100,30:GOTO 7060
7120 IF T$<>"" THEN CLS:PEN 2:PRINT"Texte en debut d'aventur
e":PEN 1:PRINT T$:PRINT:INPUT X$:IF X$="" THEN 7190 ELSE T$
=X$:GOTO 7190
7130 PEN 2:CLS:PRINT"Texte en debut d'aventure (O/N) ";;PEN
1:INPUT R$:R$=UPPER$(R$):PRINT
7140 IF R$="" THEN RETURN
7150 IF R$<>"O" AND R$<>"N" THEN SOUND 1,100,30:GOTO 7130
7160 IF R$="N" THEN 7200
7170 PEN 2:PRINT"Entrez le texte:":PEN 1
7180 INPUT T$:PRINT
7190 IF MUS$<>"" THEN CLS:PEN 2:PRINT"Nom du fichier musical
":PEN 1:PRINT MUS$:PRINT:INPUT X$:IF X$="" THEN 7250 ELSE M
US$=X$:GOTO 7250
7200 PEN 2:CLS:PRINT"Musique pendant l'aventure (O/N) ";;PEN
1:INPUT R$:R$=UPPER$(R$):PRINT
7210 IF R$="" THEN RETURN
7220 IF R$<>"O" AND R$<>"N" THEN SOUND 1,100,30:GOTO 7200
7230 IF R$="N" THEN 7250 'Pas de musique
7240 PEN 2:PRINT"Nom du fichier musical ";;PEN 1:INPUT MUS$:
PRINT
7250 PEN 2:CLS:PRINT"HELP(CTRL H) disponible (O/N) ";;PEN 1:
INPUT R$:R$=UPPER$(R$)
7260 IF R$="" THEN RETURN
7270 IF R$<>"O" AND R$<>"N" THEN SOUND 1,100,30:GOTO 7250
7280 IF R$="O" THEN HELP=1 ELSE HELP=0
7290 'Constitution du fichier aventure
7300 '
7310 AD=&6000 'Debut du fichier aventure
7311 FOR I=&6000 TO &6A00:POKE I,0:NEXT 'RAZ fichier aventur
e precedent
7320 POKE AD,SD:AD=AD+1:POKE AD,&FF:AD=AD+1 'Salle de depart
7330 IF T$="" THEN POKE AD,&FF ELSE FOR I=1 TO LEN(T$):POKE
AD,ASC(MID$(T$,I,1)):AD=AD+1:NEXT:POKE AD,&FF 'Texte de debu
t d'aventure
7340 AD=AD+1:IF MUS$="" THEN POKE AD,&FF ELSE FOR I=1 TO LEN
(MUS$):POKE AD,ASC(MID$(MUS$,I,1)):AD=AD+1:NEXT:POKE AD,&FF
'Fichier musical
7350 AD=AD+1:POKE AD,HELP:AD=AD+1:POKE AD,&FF 'Fonction HELP

7360 FOR J=1 TO 4
7370   AD=AD+1:POKE AD,N(J)
7380   FOR I=1 TO N(J)
7390     FOR K=1 TO LEN(LAB$(I,J)):AD=AD+1:POKE AD,ASC(MID$(
LAB$(I,J),K,1)):NEXT
7400   AD=AD+1:POKE AD,&FF
7410   NEXT I
7420 NEXT J
7430 'Extraction des conditions et actions
7440 AD=AD+1:POKE AD,NR 'Nombre de regles
7450 FOR I=1 TO NR
7460   FOR F=1 TO 3
7470     AD=AD+1:POKE AD,CAR(I,F)
7480   NEXT F

```



```

7490  A=99:AD=AD+1:B=LEN(C$(I))
7500  FOR J=1 TO B
7510      IF UPPER$(MID$(C$(I),J,1))="C" THEN A=VAL(RIGHT$(C$(I),B-J)):POKE AD,A 'Changements de salle
7520  NEXT J
7530  IF AD=99 THEN POKE AD,&FF
7540  A=99:AD=AD+1
7550  FOR J=1 TO LEN(C$(I))
7560      IF UPPER$(MID$(C$(I),J,1))="R" THEN A=VAL(RIGHT$(C$(I),B-J)):POKE AD,A 'Reponse
7570  NEXT J
7580  IF AD=99 THEN POKE AD,&FF
7590  A=99:AD=AD+1
7600  FOR J=1 TO LEN(C$(I))
7610      IF UPPER$(MID$(C$(I),J,1))="P" THEN A=VAL(RIGHT$(C$(I),B-J))+128:POKE AD,A 'Gestion des points
7620  NEXT J
7630  IF AD=99 THEN POKE AD,&FF
7640  A=99:AD=AD+1
7650  FOR J=1 TO LEN(C$(I))
7660      IF UPPER$(MID$(C$(I),J,1))="F" THEN POKE AD,1 'Fin de partie ou non
7670  NEXT J
7680  IF AD=99 THEN POKE AD,0
7690  NEXT I
7700  AD=AD+1:POKE AD,&FF 'Terminateur conditions
7710  '
7720  CLS:PRINT"Puis-je proceder a la sauvegarde"
7730  INPUT"de l'aventure (O/N) ";R$:R$=UPPER$(R$)
7740  IF R$<>"O" AND R$<>"N" THEN SOUND 1,100,30:GOTO 7720
7750  IF R$="N" THEN 1060
7760  PEN 2:PRINT:INPUT"Nom de l'aventure ";N$
7770  LL=AD-&6000+1
7780  SAVE N$,B,&6000,LL
7790  GOTO 1060

```

- Lignes 1060 à 1180 : Menu.
- Lignes 2030 à 2340 : Chargement d'une aventure en &H6000.
- Lignes 3000 à 3580 : Définition des salles de jeu.
- Lignes 4030 à 4100 : Définition des mots-clés.
- Lignes 5030 à 5450 : Conditions liant salles et mots-clés.
- Lignes 6030 à 6040 : Réponses de l'ordinateur.
- Lignes 7030 à 7790 : Fin d'utilisation du générateur d'aventures.
- Lignes 7030 à 7280 : Questions sur le début du jeu.
- Lignes 7290 à 7700 : Compilation de l'aventure.
- Lignes 7730 à 7780 : Sauvegarde de l'aventure.

